





© 2020 Institute of Computer Science, University
of the Philippines Los Baños



This work is licensed under the
Creative Commons
Attribution-NonCommercial-ShareAlike 4.0
International License.

To view a copy of this license, visit
<http://creativecommons.org/licenses/by-nc-sa/4.0/>
or send a letter to Creative Commons, PO Box
1866, Mountain View, CA 94042, USA.



CMSC 132, First Semester AY 2020-2021

Computer Architecture

Joseph Anthony C. Hermocilla



Sequential/State Elements



University of the Philippines
LOS BAÑOS


Clocks



University of the Philippines
LOS BAÑOS


Clocks

- Needed in sequential logic to decide when an element that contains state should be updated
- A free running signal with a **fixed cycle time or clock period**
- Cycle time is divided into two portions: clock is high or clock is low
- **Edge-triggered clocking** - all changes in state elements occur on a clock edge




Clocks (cont.)

- State elements, whose outputs change only after a clock edge, provide valid input to the combinational block
- To ensure that the values written into the state elements on the active clock edge are valid, the clock must have a long enough period



Clocks (cont.)

- Using edge-triggered methodology allows the same state element to be used as both input and output to the same combinational block



Clock in VHDL

```
1 LIBRARY ieee;  
2 USE ieee.std_logic_1164.all;  
3 ENTITY clocker_tb IS  
4 END clocker_tb;  
5 ARCHITECTURE behavior OF clocker_tb IS  
6     --100Mhz  
7     CONSTANT frequency: integer := 100e6;  
8     CONSTANT period : time := 1000 ms / frequency;  
9     SIGNAL clk : std_logic := '0';  
10 BEGIN  
11     clk <= not clk after period / 2;  
12     -- do some stuff here using clk as input  
13 END ARCHITECTURE;
```




Testing

```
$ ghdl -a clocker_tb.vhdl  
$ ghdl -e clocker_tb  
$ ghdl -r clocker_tb --vcd=clockер.vcd --stop-time=500ns  
$ gtkwave clocker.vcd
```



GTKWave output of clock



University of the Philippines
LOS BAÑOS


Memory Elements: Flip-Flops, Latches, and Registers



University of the Philippines
LOS BAÑOS

S-R latch (set-reset)


- Unclocked - no clock input




```
1  LIBRARY ieee;  
2  USE ieee.std_logic_1164.all;  
3  -----  
4  ENTITY sr_latch IS  
5      PORT (R, S: IN STD_LOGIC;  
6           Q, Q_BAR: INOUT STD_LOGIC);  
7  END sr_latch;  
8  -----  
9  ARCHITECTURE pure_logic OF sr_latch IS  
10     SIGNAL tmp: STD_LOGIC;  
11 BEGIN  
12     Q <= R NOR Q_BAR;  
13     Q_BAR <= S NOR Q;  
14 END pure_logic;
```





D latch



```
1 LIBRARY ieee;
2 USE ieee.std_logic_1164.all;
3 -----
4 ENTITY d_latch IS
5     PORT (C, D: IN STD_LOGIC;
6           Q, Q_BAR: INOUT STD_LOGIC);
7 END d_latch;
8 -----
9 ARCHITECTURE pure_logic OF d_latch IS
10 BEGIN
11     Q <= (C AND NOT D) NOR Q_BAR;
12     Q_BAR <= (D AND C) NOR Q;
13 END pure_logic;
```




D flip-flop



```
1 LIBRARY ieee;
2 USE ieee.std_logic_1164.ALL;
3 ENTITY dff IS
4     PORT( D: IN STD_LOGIC;
5           C: IN STD_LOGIC;
6           Q: INOUT STD_LOGIC;
7           Q_BAR: INOUT STD_LOGIC);
8 END dff;
9 ARCHITECTURE behavioral OF dff IS
10     COMPONENT d_latch IS
11         PORT (C, D: IN STD_LOGIC;
12              Q, Q_BAR: INOUT STD_LOGIC);
13     END COMPONENT;
14     SIGNAL u,v: STD_LOGIC;
15     SIGNAL NOTC: STD_LOGIC := NOT C;
16 BEGIN
17     u1: d_latch port map (C, D, u, v);
18     u2: d_latch port map (NOTC, u, Q, Q_BAR);
19 END behavioral;
```



D Flip-Flop Testbench



```
1  LIBRARY ieee;
2  USE ieee.std_logic_1164.all;
3  ENTITY dff_tb IS
4  END dff_tb;
5  ARCHITECTURE behavior OF dff_tb IS
6      CONSTANT frequency: integer := 100e6;
7      CONSTANT period : time := 1000 ms / frequency;
8      SIGNAL clk : std_logic := '0';
9      COMPONENT dff is
10         PORT( D: IN std_logic;
11              C: IN std_logic;
12              Q: INOUT std_logic;
13              Q_BAR: INOUT std_logic);
14     END COMPONENT;
15     SIGNAL D: std_logic := '0';
16     SIGNAL Q: std_logic := '0';
17     SIGNAL Q_BAR: std_logic;
18 BEGIN
19     clk <= not clk after period;
20     uut: dff PORT MAP (D, clk, Q, Q_BAR);
21     stim_proc: PROCESS
22     BEGIN
23         D <= '0';
24         WAIT FOR period / 2 ;
25         D <= '1';
26         WAIT FOR 20 ns;
27         D <= '0';
28         WAIT FOR 20 ns;
29         D <= '1';
30         WAIT;
31     END PROCESS;
32 END ARCHITECTURE;
```



University of the Philippines
LOS BAÑOS


Simpler Code for DFF

```
1  LIBRARY ieee;
2  USE ieee.std_logic_1164.all;
3
4  ENTITY DFF is
5  PORT( din: IN STD_LOGIC;
6        clk: IN STD_LOGIC;
7        rst: IN STD_LOGIC;
8        dout: OUT STD_LOGIC);
9  END DFF;
10
11 ARCHITECTURE behavioral of DFF is
12 BEGIN
13     PROCESS(rst,clk,din)
14     BEGIN
15         IF (rst='1') THEN
16             dout<='0';
17         ELSIF(RISING_EDGE(clk)) THEN
18             dout<= din;
19         END IF;
20     END PROCESS;
21 END behavioral;
```




Setup Time and Hold Time

- **Setup time** - minimum time that input must be valid before the clock edge
- **Hold time** - minimum time that input must be valid after the clock edge




Register Files

- A register file consists of a set of registers that can be read and written by supplying a register number to be accessed, example: 'al'




Register Files (cont..)

- Implementation of read




Register Files (cont..)

- Implementation of write



Register Files Example: with 2 1-bit registers

- With 2 1-bit registers



Register Files Example: with 2 1-bit registers (v1)

```
1  library ieee;
2  use ieee.std_logic_1164.all;
3  entity reg_file is
4      port(
5          read_n1: in std_logic;
6          read_n2: in std_logic;
7          write_n: in std_logic;
8          write_data: in std_logic := '0';
9          write: in std_logic;
10         clk: in std_logic;
11         read_data1: out std_logic;
12         read_data2: out std_logic);
13 end reg_file;
14 architecture behavioral of reg_file is
15     component dff is
16         port( D: in std_logic;
17              C: in std_logic;
18              Q: inout std_logic;
19              Q_BAR: inout std_logic);
20     end component;
21     component mux_2to1 is
22         port (A, B, S: in std_logic;
23              C: out std_logic);
24     end component;
25     component decoder_1to2 is
26         port (I: in std_logic;
27              Out0, Out1: out std_logic);
28     end component;
29     component and_gate is
30         port (A, B: in std_logic;
31              C: out std_logic);
32     end component;
```

```
33
34     signal reg0_D: std_logic;
35     signal reg0_Q: std_logic;
36     signal reg0_Q_BAR: std_logic;
37     signal reg1_D: std_logic;
38     signal reg1_Q: std_logic;
39     signal reg1_Q_BAR: std_logic;
40     signal and_0: std_logic;
41     signal and_1: std_logic;
42     signal decoder_out0: std_logic;
43     signal decoder_out1: std_logic;
44 begin
45     decoder: decoder_1to2 port map (write_n, decoder_out0, decoder_out1);
46     and0: and_gate port map (write, decoder_out0, and_0);
47     and1: and_gate port map (write, decoder_out1, and_1);
48     reg0: dff port map (write_data, and_0, reg0_Q, reg0_Q_BAR);
49     reg1: dff port map (write_data, and_1, reg1_Q, reg1_Q_BAR);
50     mux1: mux_2to1 port map (reg0_Q, reg1_Q, read_n1, read_data1);
51     mux2: mux_2to1 port map (reg0_Q, reg1_Q, read_n2, read_data2);
52 end behavioral;
```

Register Files Example: with 2 1-bit registers (v2)

```
1  LIBRARY ieee;
2  USE ieee.std_logic_1164.ALL;
3  USE IEEE.NUMERIC_STD.ALL;
4
5  ENTITY reg_file2 IS
6      PORT(
7          read_n1: IN STD_LOGIC;
8          read_n2: IN STD_LOGIC;
9          write_n: IN STD_LOGIC;
10         write_data: IN STD_LOGIC;
11         write: IN STD_LOGIC;
12         clk: IN STD_LOGIC;
13         read_data1: OUT STD_LOGIC;
14         read_data2: OUT STD_LOGIC);
15 END reg_file2;
16 ARCHITECTURE behavioral OF reg_file2 IS
17     TYPE rf_type IS ARRAY(0 to 1) OF STD_LOGIC;
18     SIGNAL registers : rf_type;
19 BEGIN
20     rf: PROCESS(clk)
21     BEGIN
22         IF RISING_EDGE(clk) THEN
23             IF (write = '1') THEN
24                 registers(TO_INTEGER(UNSIGNED>('0' & write_n))) <= write_data;
25             END IF;
26         END IF;
27         read_data1 <= registers(TO_INTEGER(UNSIGNED>('0' & read_n1)));
28         read_data2 <= registers(TO_INTEGER(UNSIGNED>('0' & read_n2)));
29     END PROCESS;
30 END behavioral;
```




Memory Elements: SRAMs and DRAMs



University of the Philippines
LOS BAÑOS


Static RAM

- Has a specific configuration in terms of the number of addressable locations and width of each addressable location
- Example: 2M x 16 SRAM
- Height = # of addressable locations, 2M
- Width = # bits per addressable unit, 16 bits




Three-state buffer


- Used as shared line which memory cells can assert
- As substitute for giant multiplexer



4 x 2 SRAM



4M × 8 SRAM as an array of 4K × 1024 arrays




Dynamic RAM

- the value kept in a cell is stored as a charge in a capacitor
- A single transistor is then used to access this stored charge, either to read the value or to overwrite the charge stored there
- use only a single transistor per bit of storage - cheaper than SRAM



DRAM

- $4\text{M} \times 1$ DRAM is built with a 2048×2048 array
- Parity codes are used to detect memory error



Finite State Machines



University of the Philippines
LOS BAÑOS


Finite State Machines

- a sequential system cannot be described with a truth table
- A **finite-state machine** has a set of states and two functions, called the next-state function and the output function
- The **set of states** corresponds to all the possible values of the internal storage
- The **next-state function** is a combinational function that, given the inputs and the current state, determines the next state of the system
- The **output function** produces a set of outputs from the current state and the inputs



Finite State Machines

- **Moore machine** - output function is dependent only on the current state
- **Mealy machine** - output function is dependent on current state and current input



Example: Traffic Light

- Outputs

- **NSLite** - When this signal is asserted, the light on the north-south road is green; when this signal is deasserted, the light on the north-south road is red
- **EWLite** - When this signal is asserted, the light on the east-west road is green when this signal is deasserted, the light on the east-west road is red

- Inputs

- **NScar**: Indicates that a car is over the detector placed in the roadbed in front of the light on the north-south road (going north or south).
- **EWcar**: Indicates that a car is over the detector placed in the roadbed in front of the light on the east-west road (going east or west).



Example: Traffic Light

- States
 - NSgreen: The traffic light is green in the north-south direction.
 - EWgreen: The traffic light is green in the east-west direction.
- Next state function (user-defined)

	Inputs		
	NScar	EWcar	Next state
NSgreen	0	0	NSgreen
NSgreen	0	1	EWgreen
NSgreen	1	0	NSgreen
NSgreen	1	1	EWgreen
EWgreen	0	0	EWgreen
EWgreen	0	1	EWgreen
EWgreen	1	0	NSgreen
EWgreen	1	1	NSgreen



Example: Traffic Light


- Output function

	Outputs	
	NSlite	EWlite
NSgreen	1	0
EWgreen	0	1




Example: Traffic Light

- Graphical Representation of State Transitions (simplified)



Example: Traffic Light

1. Assign number to states
2. $\text{NextState} = (\text{Not CurrentState AND EWCAR}) \text{ OR } (\text{CurrentState and NOT NSCAR})$
3. Output:
 - a. $\text{NSLite} = \text{NOT CurrentState}$
 - b. $\text{EWLite} = \text{CurrentState}$



VHDL Code



```
1 LIBRARY ieee;
2 USE ieee.std_logic_1164.all;
3 -----
4 ENTITY trafficlite IS
5     PORT (EWCAR, NSCAR, clk: IN STD_LOGIC;
6           EWLITE, NSLITE: OUT STD_LOGIC
7           );
8 END trafficlite;
9 -----
10 ARCHITECTURE pure_logic OF trafficlite IS
11     SIGNAL state : STD_LOGIC := '0';
12 BEGIN
13     PROCESS(clk)
14     BEGIN
15         NSLITE <= NOT state; EWLITE <= state;
16         IF (RISING_EDGE(clk)) THEN
17             CASE state IS
18                 WHEN '0' =>
19                     state <= EWCAR;
20                 WHEN '1' =>
21                     state <= NSCAR;
22                 WHEN others =>
23                     state <= '0';
24             END CASE;
25         END IF;
26     END PROCESS;
27 END pure_logic;
```



Testbench

```
1 LIBRARY ieee;
2 USE ieee.std_logic_1164.all;
3 ENTITY trafficlite_tb IS
4 END trafficlite_tb;
5 ARCHITECTURE behavior OF trafficlite_tb IS
6     --100Mhz
7     CONSTANT frequency: integer := 100e6;
8     CONSTANT period : time := 1000 ms / frequency;
9     COMPONENT trafficlite IS
10    PORT (EWCar, NSCar, clk: IN STD_LOGIC;
11          EWLite, NSLite: OUT STD_LOGIC
12    );
13    END COMPONENT;
14    SIGNAL clk : STD_LOGIC := '0';
15    SIGNAL inputs : STD_LOGIC_VECTOR(1 DOWNTO 0);
16    SIGNAL outputs : STD_LOGIC_VECTOR(1 DOWNTO 0);
17 BEGIN
18    clk <= NOT clk AFTER period / 2;
19    uut: trafficlite PORT MAP (inputs(1),inputs(0),clk,outputs(1),outputs(0));
20    stim_proc: PROCESS
21    BEGIN
22        inputs <= "00";
23        WAIT FOR period ;
24        inputs <= "10";
25        WAIT FOR period ;
26        inputs <= "01";
27        WAIT FOR period ;
28    END PROCESS;
29 END ARCHITECTURE;
```



Waveform



University of the Philippines
LOS BAÑOS

Enjoy! :)



University of the Philippines
LOS BAÑOS